RF Powers

Init

```
In[∘]:= Needs["Controls`LinearControl`"]
```

```
In[∘]:= $TextStyle = {FontFamily → "Helvetica", FontSize → 13};
```

```
In[∘]:= plotopt = Sequence @@ {GridLines → Automatic, Frame → True,
      FrameStyle → Thickness[0.0025], BaseStyle → {FontSize → 12}};
```

```
In[∘]:= plotoptn[n_Integer ? (# > 0 ∧ # < 6 &)] := Sequence @@
     {GridLines → Automatic, Frame → True, FrameStyle → Thickness[0.0025], PlotStyle →
       Take[{Purple, Brown, Darker[Green], Blue, Red}, -n], BaseStyle → {FontSize → 12}};
plotoptn[n_Integer ? (# ≤ 0 ∨ # ≥ 6 &)] := plotopt
```

```
In[∘]:= mylegend[labels_List, pos_ : Right] :=
  {Placed[LineLegend[labels, LabelStyle → {FontSize → 11},
      LegendMargins → 2, LegendFunction → (Framed[#, Background → White] &)], pos]}
```

Modulation

Calibration

alog 41889 and 41435

In[∘]:=

$$rf9cal = \frac{0.191}{dbm2v[23.6]}; (* rad/V *)$$

$$rf45cal = \frac{0.259}{dbm2v[27]}; (* rad/V *)$$

Formulae

In[∘]:=
```
dbm2v[x_] := Sqrt[10^(x/10) × 50 × 1*^-3]
rad9[x_] := rf9cal dbm2v[x]
rad45[x_] := rf45cal dbm2v[x]
pwr9[x_] := 2 BesselJ[1, rad9[x]]^2
pwr45[x_] := 2 BesselJ[1, rad45[x]]^2
pwrcr[x_, y_] := BesselJ[0, rad9[x]]^2 BesselJ[0, rad45[y]]^2
```

Calculations

Sidebands

In[∘]:=
```
rad45[24]
rad9[20.4]
pwr45[24]
pwr9[20.4]
```

Out[∘]= 0.183358

Out[∘]= 0.13214

Out[∘]= 0.0166693

Out[∘]= 0.00869241

Carrier

In[∘]:=
$$\frac{\text{pwrcr}[20.4,\ 24+3]}{\text{pwrcr}[20.4,\ 24]} - 1$$
$$\frac{\text{pwrcr}[20.4+3,\ 24]}{\text{pwrcr}[20.4,\ 24]} - 1$$

Out[∘]= -0.0166957

Out[∘]= -0.00867973

In[∘]:= $\text{Series}\left[\frac{\text{BesselJ}[0,\ \gamma]^2}{\text{BesselJ}[0,\ \gamma 1]^2},\ \{\gamma,\ 0,\ 2\},\ \{\gamma 1,\ 0,\ 2\}\right]$ // Normal // Simplify

Out[∘]= $-\frac{1}{4} \times \left(-2+\gamma^2\right) \times \left(2+\gamma 1^2\right)$

Sideband/Carrier Ratio from measured powers

c : modulation index ratio
p0: carrier power

p1: RF modulation sideband power

In[∘]:=
```
Solve[{p0 + p1 == m1, p0 + c^2 p1 == m2}, {p0, p1}]
Simplify[p1/(p0 + p1) /. %]
```

Out[∘]= $\left\{\left\{p0 \to -\frac{-c^2\, m1 + m2}{-1 + c^2},\ p1 \to -\frac{m1 - m2}{-1 + c^2}\right\}\right\}$

Out[∘]= $\left\{\frac{m1 - m2}{m1 - c^2\, m1}\right\}$

In[∘]:=
```
Solve[{BesselJ[0, γ1]^2 p0 + 2 BesselJ[1, γ1]^2 p1 == m1,
  BesselJ[0, γ2]^2 p0 + 2 BesselJ[1, γ2]^2 p1 == m2}, {p0, p1}]
Simplify[2 BesselJ[1, γ1]^2 p1/(BesselJ[0, γ1]^2 p0 + 2 BesselJ[1, γ1]^2 p1) /. %]
```

Out[∘]= $\left\{\left\{p0 \to -\frac{-m2\, \text{BesselJ}[1, \gamma1]^2 + m1\, \text{BesselJ}[1, \gamma2]^2}{\text{BesselJ}[0, \gamma2]^2\, \text{BesselJ}[1, \gamma1]^2 - \text{BesselJ}[0, \gamma1]^2\, \text{BesselJ}[1, \gamma2]^2},\ p1 \to \frac{m2\, \text{BesselJ}[0, \gamma1]^2 - m1\, \text{BesselJ}[0, \gamma2]^2}{2\left(-\text{BesselJ}[0, \gamma2]^2\, \text{BesselJ}[1, \gamma1]^2 + \text{BesselJ}[0, \gamma1]^2\, \text{BesselJ}[1, \gamma2]^2\right)}\right\}\right\}$

Out[∘]= $\left\{\frac{\left(-m2\, \text{BesselJ}[0, \gamma1]^2 + m1\, \text{BesselJ}[0, \gamma2]^2\right) \text{BesselJ}[1, \gamma1]^2}{m1\left(\text{BesselJ}[0, \gamma2]^2\, \text{BesselJ}[1, \gamma1]^2 - \text{BesselJ}[0, \gamma1]^2\, \text{BesselJ}[1, \gamma2]^2\right)}\right\}$

In[∘]:=
```
ratio[r_, c_] := (r - 1)/(c^2 - 1) (* r: measurement ratio, c: modulation index ratio *)

ratio[r_, γ1_, γ2_] := (r BesselJ[0,γ1]^2/BesselJ[0,γ2]^2 - 1)/(BesselJ[0,γ1]^2/BesselJ[0,γ2]^2 BesselJ[1,γ2]^2/BesselJ[1,γ1]^2 - 1)
```

Data

In[∘]:=
```
names = {"Input power (W)", "9MHz modulation (dBm)",
   "45MHz modulation (dBm)", "REFL_A (mW)", "REFL_B (mW)", "POP_LF",
   "POP_A_NSUM", "POP_B_NSUM", "AS_C_NSUM (W)", "OMC_A_NSUM (W)",
   "OMC_B_NSUM (W)", "TR_X_A", "TR_X_B", "TR_Y_A", "TR_Y_B"};
```

Measurements on May 12 2021, 23:44:35 UTC

alog 58946;

In[ ]:=
```
dir1 = "E:\\OneDrive\\Documents\\Mathematica\\RFpower\\13May2021\\";
(*files={"imcpwr.txt","rf9.txt","rf45.txt",
  "refl_a_lf.txt","refl_b_lf.txt","pop_lf.txt","pop_a_sum.txt",
  "pop_b_sum.txt","as_c_sum.txt","omc_a_sum.txt","omc_b_sum.txt",
  "tr_x_a_sum.txt","tr_x_b_sum.txt","tr_y_a_sum.txt","tr_y_b_sum.txt"};
paths=dir<>#&/@files; *)
file1 = "rf_pwr_210512.hdf5";
path1 = dir1 <> file1;
```

In[ ]:=
```
hdir1 = Import[path1, {"HDF5"}][[{29, 35, 32, 41, 44, 38, 11, 14, 2, 5, 8, 17, 20, 23, 26}]]
data1 = Part[#, 1 ;; 9000] & /@ Import[path1, {"HDF5", hdir1}];
```

Out[ ]=
```
{/H1:IMC-PWR_IN_OUT16/mean,
 /H1:LSC-MOD_RF9_AM_RFSET/mean, /H1:LSC-MOD_RF45_AM_RFSET/mean,
 /H1:LSC-REFL_A_LF_OUT16/mean, /H1:LSC-REFL_B_LF_OUT16/mean,
 /H1:LSC-POP_A_LF_OUT16/mean, /H1:ASC-POP_A_NSUM_OUT16/mean,
 /H1:ASC-POP_B_NSUM_OUT16/mean, /H1:ASC-AS_C_NSUM_OUT16/mean,
 /H1:ASC-OMC_A_NSUM_OUT16/mean, /H1:ASC-OMC_B_NSUM_OUT16/mean,
 /H1:ASC-X_TR_A_NSUM_OUT16/mean, /H1:ASC-X_TR_B_NSUM_OUT16/mean,
 /H1:ASC-Y_TR_A_NSUM_OUT16/mean, /H1:ASC-Y_TR_B_NSUM_OUT16/mean}
```

In[ ]:=
```
Dimensions[data1]
```

Out[ ]=
```
{15, 9000}
```

In[ ]:=
```
cursors1 = {{1650, 1940}, {2450, 2950}, {3570, 3820}, {4120, 4300},
   {4600, 4950}, {5250, 5550}, {7120, 7250}, {7700, 7900}, {8200, 8400}};
```

In[ ]:=
```
ListPlot[data1[[{2, 3}]], plotoptn[2], Joined → True,
 Prolog → {{Gray, Rectangle[{#[[1]], -1*^6}, {#[[2]], +1*^6}]} & /@ cursors1}]
```

Out[ ]=



In[∘]:=
```
ListPlot[data1[[{4, 5}]], plotoptn[2], Joined → True,
 Prolog → {{Gray, Rectangle[{#[[1]], 0}, {#[[2]], +1000000}]} & /@ cursors1}]
```

Out[∘]=



In[∘]:=
```
f[c_, d_] := Mean[d[[#[[1]] ;; #[[2]]]]] & /@ c
g[c_, d_] := Transpose[{Mean /@ c, f[c, d]}]
i1[c_, d_] := Interpolation[g[c, d][[{1, 3, 5, 7, 9}]]]
i1[d_] := i1[cursors1, d]
h1[c_, d_] := MapThread[{i1[c, d][#2[[1]]], #2[[2]], ratio[#2[[2]]/(i1[c, d][#2[[1]]]), #1]} &,
   {{1/√2, √2, 1/√2, √2}, g[c, d][[{2, 4, 6, 8}]]}]
hh1[c_, d_] := MapThread[
  {i1[c, d][#2[[1]]], #2[[2]], ratio[#2[[2]]/(i1[c, d][#2[[1]]]), #1[[1]], #1[[2]]]} &,
  {{{rad45[24], rad45[21]}, {rad45[24], rad45[27]},
    {rad9[20.4], rad9[17.4]}, {rad9[20.4], rad9[23.4]}}, g[c, d][[{2, 4, 6, 8}]]}]
```

In[∘]:=
```
g[cursors1, data1[[2]]]
```

Out[∘]=
```
{{1795, 20.4}, {2700, 20.4}, {3695, 20.4}, {4210, 20.4},
 {4775, 20.4}, {5400, 17.4}, {7185, 20.4}, {7800, 23.4}, {8300, 20.4}}
```

In[∘]:=
```
h1[cursors1, data1[[9]]]
hh1[cursors1, data1[[9]]]
```

Out[∘]=
```
{{0.319911, 0.184174, 0.84859}, {0.319382, 0.580912, 0.818865},
 {0.319548, 0.318219, 0.00832095}, {0.320838, 0.32308, 0.00698906}}
```

Out[∘]=
```
{{0.319911, 0.184174, 0.856657}, {0.319382, 0.580912, 0.839505},
 {0.319548, 0.318219, 0.016993}, {0.320838, 0.32308, 0.0157431}}
```

Table shows percentage of RF power in 45MHz (columns 2 & 3), 9MHz (columns 4 & 5) and carrier (columns 6 & 7).

Columns 2, 4 & 6 are measured by lowering the modulation depth, whereas columns 3, 5 & 7 are mea-

sured by increasing it.
First column is nominal power/dBm readback

In[∘]:=
```
res1 = hh1[cursors1, #] & /@ data1;
pwr1 = Mean /@ Map[First, res1, {2}];
frac1 = {#[[1]], #[[2]], #[[3]], #[[4]], 1 - #[[1]] - #[[3]], 1 - #[[2]] - #[[4]]} & /@ Map[Last, res1, {2}];
TableForm[MapThread[Prepend, {MapThread[Prepend, {frac1, pwr1}], names}]]
```

Out[∘]//TableForm=

Input power (W)	45.7987	0.0165969	0.0167061	0.00867367	0.008
9MHz modulation (dBm)	20.4	0.0167219	0.0167745	0.301599	0.156
45MHz modulation (dBm)	24.	0.264172	0.142365	0.00870666	0.008
REFL_A (mW)	18.0601	0.0433551	0.042632	0.056822	0.057
REFL_B (mW)	16.7815	0.0429719	0.0423067	0.056452	0.057
POP_LF	50 725.2	0.00168856	0.00219377	0.0172008	0.016
POP_A_NSUM	6045.24	0.00195324	0.00344395	0.0164113	0.016
POP_B_NSUM	5493.13	0.00241254	0.00288102	0.0168533	0.016
AS_C_NSUM (W)	0.31992	0.856657	0.839505	0.016993	0.015
OMC_A_NSUM (W)	0.304803	0.876744	0.85849	0.0163168	0.014
OMC_B_NSUM (W)	0.31393	0.905072	0.886386	0.0146934	0.013
TR_X_A	269 903.	-0.00192705	-0.00138264	-0.000581326	-0.00
TR_X_B	352 297.	-0.00172248	-0.00120213	-0.000494881	-0.00
TR_Y_A	364 211.	-0.00167934	-0.000982483	-0.000450641	-0.00
TR_Y_B	385 127.	-0.00143419	-0.000841598	-0.000372647	-0.00

In[∘]:=
```
Plot[{Interpolation[data1[[4]], InterpolationOrder → 1][x], i1[data1[[4]]][x]},
 {x, 1000, 9000}, Evaluate[plotoptn[2]],
 Prolog → {{Gray, Rectangle[{#[[1]], 0}, {#[[2]], +1000}]} & /@ cursors1}]
```

Out[∘]=



In[•]:=
```
Plot[{Interpolation[data1[[10]], InterpolationOrder → 1][x], i1[data1[[10]]][x]},
  {x, 1000, 9000}, Evaluate[plotoptn[2]],
  Prolog → {{Gray, Rectangle[{#[[1]], 0}, {#[[2]], +1000}]} & /@ cursors1}]
```

Out[•]=



In[•]:=
```
Plot[{Interpolation[data1[[9]], InterpolationOrder → 1][x], i1[data1[[9]]][x]},
  {x, 1000, 9000}, Evaluate[plotoptn[2]],
  Prolog → {{Gray, Rectangle[{#[[1]], 0}, {#[[2]], +1000}]} & /@ cursors1}, PlotRange → {0, 0.6}]
```

InterpolatingFunction: Input value {1000.16} lies outside the range of data in the interpolating function. Extrapolation will be used.

Out[•]=



In[∘]:=
```
Plot[{Interpolation[data1[[10]], InterpolationOrder → 1][x], i1[data1[[10]]][x]},
 {x, 1000, 9000}, Evaluate[plotoptn[2]],
 Prolog → {{Gray, Rectangle[{#[[1]], 0}, {#[[2]], +1000}]} & /@ cursors1}, PlotRange → {0, 0.6}]
```

InterpolatingFunction: Input value {1000.16} lies outside the range of data in the interpolating function. Extrapolation will be used.

Out[∘]=


In[∘]:=
```
Plot[{Interpolation[data1[[11]], InterpolationOrder → 1][x], i1[data1[[11]]][x]},
 {x, 1000, 9000}, Evaluate[plotoptn[2]],
 Prolog → {{Gray, Rectangle[{#[[1]], 0}, {#[[2]], +1000}]} & /@ cursors1}, PlotRange → {0, 0.6}]
```

InterpolatingFunction: Input value {1000.16} lies outside the range of data in the interpolating function. Extrapolation will be used.

Out[∘]=


Measurements on April 17 2019, 00:30:00 UTC

alog 48595

In[∘]:=
```
dir2 = "E:\\OneDrive\\Documents\\Mathematica\\RFpower\\17Apr2019\\";
file2 = "rf_pwr_190417.hdf5";
path2 = dir2 <> file2;
```

In[∘]:=
```
hdir2 = Import[path2, {"HDF5"}][[{10, 12, 11, 14, 15, 13, 4, 5, 1, 2, 3, 6, 7, 8, 9}]]
data2 = Part[#, 12500 ;; 22500] & /@ Import[path2, {"HDF5", hdir2}]
   {1, 1, 1, 1, 1, 1, 1, 1, 0.5, 0.16, 0.16, 1, 1, 1, 1};
```

Out[∘]=
```
{/H1:IMC-PWR_IN_OUT16/raw, /H1:LSC-MOD_RF9_AM_RFSET/raw,
 /H1:LSC-MOD_RF45_AM_RFSET/raw, /H1:LSC-REFL_A_LF_OUT16/raw,
 /H1:LSC-REFL_B_LF_OUT16/raw, /H1:LSC-POP_A_LF_OUT16/raw, /H1:ASC-POP_A_NSUM_OUT16/raw,
 /H1:ASC-POP_B_NSUM_OUT16/raw, /H1:ASC-AS_C_NSUM_OUT16/raw,
 /H1:ASC-OMC_A_NSUM_OUT16/raw, /H1:ASC-OMC_B_NSUM_OUT16/raw,
 /H1:ASC-X_TR_A_NSUM_OUT16/raw, /H1:ASC-X_TR_B_NSUM_OUT16/raw,
 /H1:ASC-Y_TR_A_NSUM_OUT16/raw, /H1:ASC-Y_TR_B_NSUM_OUT16/raw}
```

In[∘]:=
```
Dimensions[data2]
```

Out[∘]=
```
{15, 10001}
```

In[∘]:=
```
cursors2 = {{2700, 3700}, {4050, 4426}, {4750, 5650}, {6110, 6370}, {6750, 7750}};
```

In[∘]:=
```
ListPlot[data2[[{2, 3}]], plotoptn[2], Joined → True,
 Prolog → {{Gray, Rectangle[{#[[1]], -1*^6}, {#[[2]], +1*^6}]} & /@ cursors2}, PlotRange → All]
```

Out[∘]=



In[∘]:=
```
i2[c_, d_] := Interpolation[g[c, d][[{1, 3, 5}]], InterpolationOrder → 2]
i2[d_] := i2[cursors2, d]
h2[c_, d_] := MapThread[{i2[c, d][#2[[1]]], #2[[2]], ratio[#2[[2]] / i2[c, d][#2[[1]]], #1]} &,
   {{10^(2/20), √2}, g[c, d][[{2, 4}]]}]
hh2[c_, d_] := MapThread[
   {i2[c, d][#2[[1]]], #2[[2]], ratio[#2[[2]] / i2[c, d][#2[[1]]], #1[[1]], #1[[2]]]} &,
   {{{rad45[24], rad45[26]}, {rad9[20.4], rad9[23.4]}}, g[c, d][[{2, 4}]]}]
```

In[∘]:=
```
g[cursors2, data2[[2]]]
```

Out[∘]=
```
{{3200, 20.4}, {4238, 20.4}, {5200, 20.4}, {6240, 23.4}, {7250, 20.4}}
```

In[ ]:=
```
h2[cursors2, data2⟦4⟧]
hh2[cursors2, data2⟦4⟧]
```

Out[ ]= {{5.28774, 5.66894, 0.123258}, {5.29688, 5.44126, 0.0272578}}

Out[ ]= {{5.28774, 5.66894, 0.139588}, {5.29688, 5.44126, 0.0361078}}

Table shows percentage of RF power in 45MHz (column 2), 9MHz (column 3) and carrier (column 4)
First column is nominal power/dBm readback

In[ ]:=
```
res2 = hh2[cursors2, #] & /@ data2;
pwr2 = Mean /@ Map[First, res2, {2}];
frac2 = {#⟦1⟧, #⟦2⟧, 1 - #⟦1⟧ - #⟦2⟧} & /@ Map[Last, res2, {2}];
TableForm[MapThread[Prepend, {MapThread[Prepend, {frac2, pwr2}], names}]]
```

Out[ ]//TableForm=

Input power (W)	35.4426	0.0166753	0.00881129	0.974513
9MHz modulation (dBm)	20.4	0.01676	0.156477	0.826763
45MHz modulation (dBm)	24.	0.158738	0.00872087	0.832541
REFL_A (mW)	5.29231	0.139588	0.0361078	0.824305
REFL_B (mW)	4.89711	0.138672	0.0363148	0.825013
POP_LF	32137.6	0.00360145	0.0116421	0.984756
POP_A_NSUM	3870.93	0.0038907	0.0115728	0.984537
POP_B_NSUM	3555.27	0.00403275	0.0115129	0.984454
AS_C_NSUM (W)	0.245145	0.697172	0.0884979	0.21433
OMC_A_NSUM (W)	0.226191	0.753346	0.0796739	0.16698
OMC_B_NSUM (W)	0.220625	0.801657	0.0622945	0.136049
TR_X_A	172621.	-0.000472253	-0.000619759	1.00109
TR_X_B	226431.	-0.000214629	-0.000496668	1.00071
TR_Y_A	247014.	-0.000484864	-0.000561334	1.00105
TR_Y_B	262219.	-0.00037168	-0.000521747	1.00089

In[ ]:=
```
Plot[{Interpolation[data2⟦4⟧, InterpolationOrder → 1][x], i2[data2⟦4⟧][x]},
 {x, 2000, 9000}, Evaluate[plotoptn[2]], PlotRange → All,
 Prolog → {{Gray, Rectangle[{#⟦1⟧, 0}, {#⟦2⟧, +1*^6}]} & /@ cursors2}]
```

Out[ ]=



In[∘]:=
```
Plot[{Interpolation[data2[[9]], InterpolationOrder → 1][x], i2[data2[[9]]][x]},
 {x, 2000, 9000}, Evaluate[plotoptn[2]], PlotRange → All,
 Prolog → {{Gray, Rectangle[{#[[1]], 0}, {#[[2]], +1*^4}]} & /@ cursors2}]
```

Out[∘]=



Compare Measurements

Ratios: New/Old

Table shows total power ratio between new and old (column 1), 45 MHz power ratio (column 2), 9 MHz power ratio (column 3), and carrier power ratio (column 4)

In[∘]:=
```
pwrratio = Drop[pwr1 / pwr2, {2, 3}];
sb1pwr = Part[#, {2, 4, 6}] & /@ (pwr1 frac1);
sb2pwr = pwr2 frac2;
sbratio = Transpose[Drop[sb1pwr / sb2pwr, {2, 3}]];
TableForm[Transpose[{Drop[names, {2, 3}], pwrratio, Sequence @@ sbratio}]]
```

Out[∘]//TableForm=

Input power (W)	1.29219	1.29459	1.27187	1.29234
REFL_A (mW)	3.41251	1.04223	5.43383	3.72535
REFL_B (mW)	3.42682	1.04547	5.37938	3.74114
POP_LF	1.57838	0.961443	2.29514	1.57216
POP_A_NSUM	1.56171	1.38238	2.18417	1.5551
POP_B_NSUM	1.54507	1.1038	2.21938	1.53899
AS_C_NSUM (W)	1.30502	1.57145	0.232152	0.881372
OMC_A_NSUM (W)	1.34755	1.53562	0.251898	1.02181
OMC_B_NSUM (W)	1.42291	1.5733	0.308318	1.0471
TR_X_A	1.56356	4.57774	2.30264	1.56544
TR_X_B	1.55586	8.71437	2.5713	1.5579
TR_Y_A	1.47445	2.9877	1.53764	1.47522
TR_Y_B	1.46873	3.32564	1.57271	1.46947

Input-Power Normalized Ratios: New/Old

The total power in reflection has increased by 2.65 (shot noise x1.6), whereas the 9 MHz sidebands have increase by ~4.18 (signal x2.05). The carrier gain in the PRC is also 1.21 times higher (signal 1.1), yielding an improvement of the SNR of about 1.4 at equivalent input power, or ~1.6 with the increased input power.

At the AS port the total power stayed about the same for equivalent input power with carrier contribution decreasing by ~0.7-0.8, the 45 MHz contribution increasing by ~1.2, and the 9 MHz reducing by a factor of ~4-5.5. With the increase of the 9 MHz sideband gain of ~1.7 in the PRC, the 9 MHz contrast defect has improved by roughly 7-10.

From the estimated 320 mW at the AS port, 269 mW are 45 MHz sidebands, 32-46 mW are carrier and 4-5 mW are 9 MHz sidebands. With 23.3mW of the carrier in the TEM00, we end up with 9-23mW in carrier HOM. This is significantly better than with the old ITM, where the carrier TEM00 and HOM were about the same at a lower input power. Estimating 2.2 kW of carrier light at the beamsplitter, the carrier contrast defect is now about 13-32 ppm, where we had 20-66 pm previously. However 87 ppm are reported in alog 56590. The uncertainty is due to the rather inconsistent measurements coming from ASC-AS_C, OMC_A and OMC_B.

Table shows input-power normalized power ratio between new and old (column 1), 45 MHz power ratio (column 2), 9 MHz power ratio (column 3), and carrier power ratio (column 4)

In[∘]:=
```
pwrdoubleratio = pwrratio / pwrratio[[1]];
sbdoubleratio = sbratio / pwrratio[[1]];
TableForm[Transpose[{Drop[names, {2, 3}], pwrdoubleratio, Sequence @@ sbdoubleratio}]]
```

Out[∘]//TableForm=

Input power (W)	1.	1.00185	0.984271	1.00011
REFL_A (mW)	2.64087	0.806558	4.20513	2.88297
REFL_B (mW)	2.65194	0.809063	4.16299	2.89519
POP_LF	1.22147	0.74404	1.77616	1.21666
POP_A_NSUM	1.20857	1.0698	1.69028	1.20346
POP_B_NSUM	1.19569	0.854211	1.71753	1.19099
AS_C_NSUM (W)	1.00993	1.21611	0.179658	0.682075
OMC_A_NSUM (W)	1.04284	1.18838	0.194938	0.790756
OMC_B_NSUM (W)	1.10116	1.21754	0.238601	0.810324
TR_X_A	1.21001	3.54261	1.78197	1.21146
TR_X_B	1.20405	6.74386	1.98988	1.20563
TR_Y_A	1.14105	2.31211	1.18995	1.14164
TR_Y_B	1.13661	2.57364	1.21709	1.13719

Power Levels New

alog 46386
Input efficiency from PSL to PRM: 88.8%
Loss in REFL path 28.4% with 4 50:50 and 1 90:10 beamsplitter. Overall efficiency: 1/223.
Estimate that about 10% of the incident carrier power is at the reflected port, ~25% of the 45MHz sidebands, and ~65% of the 9 MHz sidebands.

Table shows total power (column 1), 45 MHz power (column 2), 9 MHz power (column 3), and carrier power (column 4)

In[ ]:=
```
TableForm[Transpose[
  {Drop[names, {2, 3}], Drop[pwr1, {2, 3}], Sequence @@ Transpose[Drop[sb1pwr, {2, 3}]]}]]
```

Out[ ]//TableForm=

Input power (W)	45.7987	0.76512	0.397198	44.6364
REFL_A (mW)	18.0601	0.769936	1.03837	16.2517
REFL_B (mW)	16.7815	0.709969	0.956657	15.1149
POP_LF	50 725.2	111.279	858.722	49 755.2
POP_A_NSUM	6045.24	20.8195	97.8451	5926.58
POP_B_NSUM	5493.13	15.8258	90.8429	5386.46
AS_C_NSUM (W)	0.31992	0.268574	0.00503651	0.0463091
OMC_A_NSUM (W)	0.304803	0.26167	0.00453958	0.038593
OMC_B_NSUM (W)	0.31393	0.278263	0.00423745	0.0314294
TR_X_A	269 903.	-373.18	-246.344	270 523.
TR_X_B	352 297.	-423.507	-289.172	353 009.
TR_Y_A	364 211.	-357.831	-213.205	364 782.
TR_Y_B	385 127.	-324.122	-215.166	385 667.

Power Levels Old

Table shows total power (column 1), 45 MHz power (column 2), 9 MHz power (column 3), and carrier power (column 4)

In[ ]:=
```
TableForm[Transpose[
  {Drop[names, {2, 3}], Drop[pwr2, {2, 3}], Sequence @@ Transpose[Drop[sb2pwr, {2, 3}]]}]]
```

Out[ ]//TableForm=

Input power (W)	35.4426	0.591015	0.312295	34.5393
REFL_A (mW)	5.29231	0.73874	0.191094	4.36247
REFL_B (mW)	4.89711	0.679094	0.177838	4.04018
POP_LF	32 137.6	115.742	374.148	31 647.7
POP_A_NSUM	3870.93	15.0606	44.7973	3811.07
POP_B_NSUM	3555.27	14.3375	40.9316	3500.
AS_C_NSUM (W)	0.245145	0.170909	0.0216949	0.0525421
OMC_A_NSUM (W)	0.226191	0.1704	0.0180215	0.0377693
OMC_B_NSUM (W)	0.220625	0.176866	0.0137437	0.0300158
TR_X_A	172 621.	-81.5206	-106.983	172 809.
TR_X_B	226 431.	-48.5988	-112.461	226 592.
TR_Y_A	247 014.	-119.768	-138.657	247 272.
TR_Y_B	262 219.	-97.4616	-136.812	262 453.